Systemplanering 2014

Lecture 9, L9: **Short-time planning of systems with hydro power and thermal power**

Systemplanering 2014

- Chapters
  - 5.2.5,
  - 5.3.4,
  - 5.4,
  - Appendix B
- Content:
  - Planning problems
  - Dual variables
  - Home Assignment III

Dual Variables (1/4)

- Repetition
  - LP, standard form

$$\min_x c^T x$$

  - subject to $Ax = b$

$$x \geq 0$$

  - Each solution (for most solvers) results in dual variables
    - One dual variable for each constraint
    - Interpretation: objective's sensitivity against RHS alterations

Dual Variables (2/4)

At optimum:

- $\lambda_1 > 0$ (active)
- $\lambda_2 > 0$ (active)
- $\lambda_3 = 0$ (inactive)





Dual Variables (3/4)

- Hydrological constraints
  - $M_{1,t} - M_{1,t-1} + S_{1,t} + Q_{1,t} = V_1, \text{ t} \in \{1,2, \dots, 6\}$
  - $M_{2,t} - M_{2,t-1} + S_{2,t} + Q_{2,t} - S_{1,t} - Q_{1,t} = V_2, \text{ t} \in \{1,2, \dots, 6\}$
  - Expressed in units HE
- Objective is some monetary unit (SEK, $, £, €, ¤)
- Increased V, more water, more money
- Dual variables correspond to water values
  - Unit: [¤/HE]

Dual Variables (4/4)

- Constraint for contracted load
  - $\sum_{i,j} \mu_{ij} Q_{ijt} = D_t, \quad t \in \{1,2,\ldots,6\}$
  - Unit MWh
- Objective still in monetary units
- Increased D,
  - Increased production cost, or
  - Less water left
  - Objective decreases
- Dual variables reflects marginal production costs
  - Unit: [¤/MWh]

Dual Variables – Example (1/2)

- 6 hours hydro planning
  - 2 power plants
  - Parts of the load contracted beforehand
- Resulting dual variable values at optimum
  - 82.4304 hydrologic constraints plant 1, hours 1-6
  - 42.6585 hydrologic constraints plant 1, hours 1-6
  - -190.0389 load balance, hours 1-5,6
  - -200.0410 load balance, hour 4

Dual Variables – Example (2/2)

- Resulting dual variable values at optimum
  - 82.4304 hydrologic constraints plant 1, hours 1-6
  - 42.6585 hydrologic constraints plant 2, hours 1-6
  - -190.0389 load balance, hours 1-5,6
  - -200.0410 load balance, hour 4
- Questions:
  - What about the structure of the system?
  - Someone wants to contract another 5 MWh, hour 2
  - Lowest profitable price?

Home Assignment III

- Short-time planning
  - Hydro power, and
  - Thermal Power
- Recommended software, GAMS



GAMS (1/5)

- Algebraic modelling software
  - Solving optimization problems, mainly
  - Also CNS, and various KKT things
- Typical structure of a GAMS-model
  - Define sets
  - Define parameters
  - Define variables
  - Continues …

GAMS (2/5)

- Typical structure of a GAMS-model
  - Continued …
    - Variable types
      - Free (default)
      - Positive (i.e. nonnegative)
      - Negative (i.e. nonpositive)
      - Continues …

GAMS (3/5)

- Typical structure of a GAMS-model
  - Define variables
    - Variable types
      - Continued …
      - Integer
      - Binary (special case of integer)
      - Semi continuous (e.g. minimal production)
      - Semi integer
      - SOS1, SOS2, …
    - Continues …

GAMS (4/5)

- Typical structure of a GAMS-model
  - Define variables
    - Continued …
    - Duals implicitly defined by equations
    - Variable bounds (better than constraints)
  - Continues …

GAMS (5/5)

- Typical structure of a GAMS-model
  - Continued …
  - Equations (i.e. constraints)
    - Declare
    - Define
  - Define model(s)
  - Call for solver
    - Define problem type
    - Which model to use



Example 5.6 (1/8)



- Given
  - Two hydro power plants
  - Contracted load the 6 following hours
    - 90, 98, 104, 112, 100, 80 MWh
    - During this period, no further production!
  - Reservoirs 50% filled before first hour
  - Water saved after planning period
    - Used for most efficient production
    - Sold for 185 SEK/MWh
  - Water transport times neglected

Example 5.6 (2/8)



Power plant	Maximal reservoir content $[\mathrm{m}^3]$	Marginal production equivalents $[\frac{\mathrm{MWh}}{\mathrm{HE}}]$, Q in [HE]	Annual average water inflow, w_i $[\frac{\mathrm{m}^3}{s}]$
Degerforsen	$5 \cdot 10^6$	$0.209, 0 \leq Q \leq 225$ $0.199, 225 \leq Q \leq 300$	163
Edensforsen	$4 \cdot 10^6$	$0.236, 0 \leq Q \leq 202.5$ $0.224, 202.5 \leq Q \leq 270$	164

Example 5.6 (3/8)





- Model creation
  - Indices
    - Power Plants, $i \in \{1,2\}$
    - Segments, $j \in \{1,2\}$
    - Hours, $t \in \{1,2,\ldots,6\}$
  - Continued …



Example 5.6 (4/8)



- Model creation
  - Continues …
  - Parameters
    - $\mu_{i,j},\ \bar{Q}_{i,j},\ \bar{M}_i,\ \lambda_f,\ w_i,\ D_t,\ M_{i,0}$
    - $V_1 = w_1$
    - $V_2 = w_2 - w_1$
  - Continued …

Example 5.6 (5/8)





- Model creation
  - Continues …
  - Variables
    - Discharge, $Q_{i,j,t}$, plant i, segment j, hour t.
    - Spillage, $S_{i,t}$, plant i, hour t
    - Reservoir content, $M_{i,t}$, plant i, end of hour t
  - Continued …



Example 5.6 (6/8)



- Model creation
    - Variables
    - Continues …
        - Variable limits
        - $0 \leq Q_{i,j,t} \leq \bar{Q}_{i,j}$
        - $0 \leq S_{i,t}$
        - $0 \leq M_{i,t} \leq \bar{M}_i$
- Continued …



Example 5.6 (7/8)



- Model creation
  - Continues …
  - Objective, $\max z$
  - Subject to constraints:
    - $z = \lambda_f \left((\mu_{1,1} + \mu_{2,1}) M_{1,6}) + \mu_{2,1} M_{2,6} \right)$
    - Continued …



Example 5.6 (8/8)



- Model creation
  - Subject to constraints:
    - Continues …
    - $M_{1,t} - M_{1,t-1} + \sum_{j=1}^{2} Q_{1,j,t} + S_{1,t} = V_1$
    - $M_{2,t} - M_{2,t-1} + \sum_{j=1}^{2} Q_{2,j,t} + S_{2,t} - \sum_{j=1}^{2} Q_{1,j,t} - S_{1,t} = V_2$
    - $\sum_{i=1,j=1}^{i=1,j=2} \mu_{i,j} Q_{i,j,t} = D_t$

GAMS Code Example (1/14)






```
Sets
    i Plant          /Degerforsen, Edensforsen/
    j segment        /segment1*segment2/
    t hour           /hour1*hour6/
    ;

Parameters
    Hmax(i)          installed power in plant i
                     /Degerforsen          62
                     Edensforsen           63/
    Qtotmax(i)       maximal discharge in plant i
                     /Degerforsen          300
                     Edensforsen           270/
    Mmax(i)          maximal storage in reservoir i
                     /Degerforsen          5e6
                     Edensforsen           4e6/
    lambdaf          expected future electricity price /185/
(continues …)
```

GAMS Code Example (2/14)






```
Parameters
    …
(continued …)
          w(i)      mean annual flow at plant i
                    /Degerforsen       163
                    Edensforsen        164/
          D(t)      contracted load hour t
                    /timme1 90, timme2 98, timme3 104,
                    timme4 112, timme5 100, timme6 80/
          Mstart(i) initial content in reservoir i
          V(i)      mean annual local flow at reservoir i
          Qmax(i,j) maximal discharge at plant i, segment j
          my(i,j)   marginal production equivalent for plant i, segment j
          ;
(continues …)
```

Warning!

GAMS Code Example (3/14)






```
Parameters
    …
(continued …)
          w(i)       mean annual flow at plant i
                     /Degerforsen        163
                     Edensforsen         164/
          D(t)       contracted load hour t
                     /timme1 90, timme2 98, timme3 104,
                     timme4 112, timme5 100, timme6 80/
          Mstart(i)  initial content in reservoir i
          V(i)       mean annual local flow at reservoir i
          Qmax(i,j)  "maximal discharge at plant i, segment j"
          my(i,j)    "marginal production equivalent for plant i, segment j"
          ;
(continues …)
```

Commas
=>
quotation marks
needed!

GAMS Code Example (4/14)






```
Parameters
    …
(continued …)
         w(i)      mean annual flow at plant i
                   /Degerforsen        163
                   Edensforsen         164/
         D(t)      contracted load hour t
                   /timme1 90, timme2 98, timme3 104,
                   timme4 112, timme5 100, timme6 80/
         Mstart(i) initial content in reservoir i
         V(i)      mean annual local flow at reservoir i
         Qmax(i,j) "maximal discharge at plant i, segment j"
         my(i,j)   "marginal production equivalent for plant i, segment j"
         ;
(continues …)
```



GAMS Code Example (5/14)




```
(continued …)
Mmax(i)                    = Mmax(i)/3600;
Mstart(i)                  = 0.5*Mmax(i);
V(i)                       = w(i) - w(i - 1);
Qmax(i,"segment1")         = 0.75*Qtotmax(i);
Qmax(i,"segment2")         = Qtotmax(i) - Qmax(i,"segment1");
my(i,"segment1")           = max(i)/(Qmax(i,"segment1") +
        0.95*Qmax(i,"segment2"));
my(i,"segment2")           = 0.95*my(i,"segment1");

(continues …)
```



GAMS Code Example (6/14)




```
(continued …)
Positive variables
        Q(i,j,t) "discharge in plant i, segment j, hour t"
        S(i,t)   "spillage from reservoir i, hour t"
        M(i,t)   content in reservoir i at the end of hour t
        ;
Free variable
        z        the value of stored water
        ;

(continues …)
```



GAMS Code Example (7/14)




```
(continued …)
Q.up(i,j,t)                  = Qmax(i,j);
M.up(i,t)                    = Mmax(i);
Q.lo(i,j,t)                  = 0;
M.lo(i,t)                    = 0;
```

Not needed when defined positive

```
(continues …)
```



GAMS Code Example (8/14)




```
(continued …)
Q.up(i,j,t)                 = Qmax(i,j);
M.up(i,t)                   = Mmax(i);
```

etc...

Alternatively:

```
loop(t,
         Q.up(i,j,t)        = Qmax(i,j);
         M.up(i,t)          = Mmax(i)
         );
(continues …)
```

GAMS Code Example (9/14)






```
(continued …)
Equation
           malfkn          målfunktion;
malfkn..        z =e=          lambdaf*((my(”Degerforsen”,”segment1”)
            + my(”Edensforsen”,”segment1”))*M(”Degerforsen”,”timme6”) +
            my(”Edensforsen”,”segment1”)*M(”Edensforsen”,”timme6”));

(continues …)
```

- Sets, variables, parameters, and equations
- can be stated
  - **one and one**
  - or grouped



GAMS Code Example (10/14)




```
(continued …)
Equations
         hydbal(i,t)      "hydrological balance for plant i, hour t"
         loadbal(t)       load balance for hour t
         ;
hydbal(i,t) ..    M(i,t)   =e= M(i,t - 1) + Mstart(i)$(ord(t) = 1) –
         sum(j,Q(i,j,t)) – S(i,t) + sum(j,Q(i - 1,j,t)) + S(i - 1,t) -
         V(i);
loadbal(t) ..     sum((i,j),my(i,j)*Q(i,j,t))
                           =e= D(t)
         ;
(continues …)
```

- Sets, variables, parameters, and equations
- can be stated
  - one and one
  - **or grouped**



GAMS Code Example (11/14)




```
(continued …)
model vattenkraftsplanering /all/;
```

Alternatively:

```
model vattenkraftsplanering / malfkn, hydbal, loadbal /;
```

```
(continues …)
```

GAMS Code Example (12/14)






```
(continued …)
model vattenkraftsplanering /all/;



Option lp = bdmlp;

solve vattenkraftsplanering  using lp maximizing z;

(continues …)
```



GAMS Code Example (13/14)



(continued …)

Postprocessing the solution for a structured output

```
Parameters
         Qtot(i,t)  "total discharge in plant i, hour t"
         H(i,t)     "electricity production in plant i, hour t"
         ;

loop((i,t),
         Qtot(i,t)  = sum(j,Q.L(i,j,t));
         H(i,t)     = sum(j,my(i,j)*Q.L(i,j,t))
         );

display M.L, Q.L, Qtot, H, S.L;
display hydbal.M, loadbal.M;
```

Complicated alternative



GAMS Code Example (14/14)



(continued …)
Postprocessing the solution for a structured output

```
Parameters
         Qtot(i,t)  "total discharge in plant i, hour t"
         H(i,t)     "electricity production in plant i, hour t"
         ;


Qtot(i,t)           = sum(j,Q.L(i,j,t));
H(i,t)              = sum(j,my(i,j)*Q.L(i,j,t))


display M.L, Q.L, Qtot, H, S.L;
display hydbal.M, loadbal.M;
```

Simple alternative

Go Home

- The show is over!